



*National Aeronautics and Space Administration
Goddard Earth Science
Data Information and Services Center (GES DISC)*

README Document for the Nimbus-4 Infrared Interferometer Spectrometer (IRIS) Level 1 Radiance Data

IRISN4RAD

Last Revised 03/12/2015

Goddard Earth Sciences Data and Information Services Center (GES DISC)
<http://disc.gsfc.nasa.gov>
NASA Goddard Space Flight Center
Code 610.2
Greenbelt, MD 20771 USA

Prepared By:

James E. Johnson

03/12/2015

Name
GES DISC
GSFC Code 610.2

Date

Reviewed By:

Name

mm/dd/yyyy

Name
GSFC Code xxx

Date

Name

mm/dd/yyyy

Name
GSFC Code xxx

Date

Goddard Space Flight Center
Greenbelt, Maryland

Revision History

<i>Revision Date</i>	<i>Changes</i>	<i>Author</i>
03/12/2015	Original	James E. Johnson

Table of Contents

1. Introduction	5
1.1 Data Product Description	5
1.1.1 The Infrared Interferometer Spectrometer	5
1.1.2 Nimbus-4 Overview.....	6
1.2 Algorithm Background.....	6
1.3 Data Disclaimer	6
2. Data Organization	8
2.1 File Naming Convention	8
2.2 File Format and Structure	8
2.3 Key Science Data Fields.....	9
3. Data Contents	10
3.1 Data Records	10
3.2 Metadata	16
4. Reading the Data.....	18
5. Data Services.....	19
5.1 Reverb	19
5.2 FTP	19
6. More Information	20
6.1 Web Resources.....	20
6.2 Point of Contact.....	20
6.3 References.....	20
7. Appendices.....	21
Acknowledgements.....	21
Acronyms	21
Problem Files.....	22
FORTTRAN Code.....	23

1. Introduction

This document provides basic information on using the Nimbus-4 Infrared Interferometer Spectrometer (IRIS) Level-1 Radiance Data product.

1.1 Data Product Description

The Nimbus-4 Infrared Interferometer Spectrometer (IRIS) Level-1 Radiance Data product contains the calibrated radiances of thermal emissions from the earth and atmosphere for wave numbers between 400 and 1600 cm^{-1} with a nominal resolution of 2.81 cm^{-1} . Each file typically contains one full day's worth of data. Spatial coverage is global. The data are available from April 9, 1970 (day of year 99) through January 30, 1971 (day of year 30).

This product was previously available from the NASA National Space Science Data Center (NSSDC) under the name Infrared Interferometer Spectrometer (IRIS) Radiance Data with the identifier ESAD-00022 (old id 70-025A-03A).

1.1.1 The Infrared Interferometer Spectrometer

The Infrared Interferometer Spectrometer (IRIS) was designed to provide information on the vertical structure of the atmosphere and on the emissive properties of the earth's surface by measuring the surface and atmospheric radiation in the 6.25- to 25- μm (400 to 1600 cm^{-1}) range using a modified Michelson interferometer. Radiation from a cone of the atmosphere, whose base on the surface of the earth was a circle about 94 km in diameter for a nominal satellite altitude of approximately 1100 km, was received and reflected by a mirror. The reflected radiation was split into two approximately equal beams by a beamsplitter. After reflection on a fixed and moving mirror, respectively, the two beams interfered with each other with a phase difference proportional to the optical path difference between both beams. The moving mirror traveled about 3.6 mm in 13 seconds to give an output signal from the bolometer. This signal, an interferogram, was recorded on tape. The interferograms were transmitted to a ground receiving station, where a Fourier transform was performed to produce a thermal emission spectrum of the earth. From these spectra, vertical profiles of temperature, water vapor, and ozone can be derived, as well as other parameters of meteorological interest. The instrument had a field of view of 5° and a spectral resolution of less than 0.4 micrometer (nominally 1.4 reciprocal centimeters).

The Nimbus-4 IRIS mission was a follow-on to a similar IRIS flown on the previous Nimbus-3 satellite. The experiment was successful in spite of a transmission conflict with the Real-Time Transmission System (RTTS) that resulted in some periods of lost data after November 28, 1970. The IRIS experiment was turned off on January 25, 1972 to conserve spacecraft power.

The principal investigator for the IRIS experiment was Dr. Rudolf A. Hanel.

1.1.2 Nimbus-4 Overview

The Nimbus-4 satellite was successfully launched on April 8, 1970. The spacecraft included nine experiments: (1) an Image Dissector Camera System (IDCS) for providing daytime cloud cover pictures, both in real-time and recorded modes (2) a Temperature-Humidity Infrared Radiometer (THIR) for measuring daytime and nighttime surface and cloudtop temperatures, as well as the water vapor content of the upper atmosphere, (3) an Infrared Interferometer Spectrometer (IRIS) for measuring the emission spectra of the earth/atmosphere system, (4) a Satellite Infrared Spectrometer (SIRS) for determining the vertical profiles of temperature and water vapor in the atmosphere, (5) a Monitor of Ultraviolet Solar Energy (MUSE) for detecting solar UV radiation, (6) a Backscatter Ultraviolet (BUV) detector for monitoring the vertical distribution and total amount of atmospheric ozone on a global scale, (7) a Filter Wedge Spectrometer (FWS) for accurate measurement of IR radiance as a function of wavelength from the earth/atmosphere system, (8) a Selective Chopper Radiometer (SCR) for determining the temperatures of six successive 10-km layers in the atmosphere from absorption measurements in the 15-micrometer CO₂ band, and (9) an Interrogation, Recording, and Location System (IRLS) for locating, interrogating, recording, and retransmitting meteorological and geophysical data from remote collection stations.

The orbit of the satellite can be characterized by the following:

- circular orbit at 1100 km
- inclination of 80 degrees
- period of an orbit is about 107 minutes
- orbits cross the equator at 26 degrees of longitude separation
- sun-synchronous

1.2 Algorithm Background

The Nimbus-4 IRIS data were generated from the spacecraft telemetry, attitude and orbital data. The data were originally processed on IBM 360 computers using a 32-bit architecture. The data processing consists of four steps: (1) checking the consistency of the input raw data tapes and processing of housekeeping information, (2) Fourier transform each interferogram, (3) Application of calibration procedure, (4) production of archival tape with calibrated radiances and housekeeping information and orbital parameters. Further information on the IRIS data processing can be found in the Nimbus-4 Users' Guide Section 4.

1.3 Data Disclaimer

The data should be used care and one should first read the Nimbus-4 User's Guide, section 4 describing the IRIS experiment. Users should cite this data product in their research.

2. Data Organization

The Nimbus-4 Infrared Interferometer Spectrometer Level-1 Radiance Data spans the time period from April 9, 1970 to January 30, 1971. Each file typically contains about one day's worth of data.

2.1 File Naming Convention

The data product files are named according to the following convention:

<Instrument>-<Platform>_<Date>_<OrbitBegin>-<OrbitEnd>.<Suffix>

where:

- Instrument = name of the instrument (always IRIS)
- Platform = name of the platform or satellite (always Nimbus4)
- Date = Data start date and time in UTC in format <YYYY>m<MMDD>t<hhmm> where
 1. YYYY = 4 digit year (1970 or 1971)
 2. MM = 2 digit month (01-12)
 3. DD = 2 digit day of month (01-31)
 4. hh = 2 digit hour of day (01-23)
 5. mm = 2 digit minute of hour (01-59)
- OrbitBegin = first orbit when the data were collected (preceded by the letter 'o')
- OrbitEnd = last orbit when the data were collected
- Suffix = the file format (always dat, indicating binary data)

File name example: IRIS-Nimbus4_1970m0409t1647_o19-22.TAP

2.2 File Format and Structure

The data are stored as they were originally written in IBM binary (big-endian) record oriented structured files. The files were written on the original 9-track tapes using a blocked FORTRAN format with a size that is an increment of 3572 bytes (893 words). The data archived have the tape FORTRAN leading and trailing block record size words removed, but include 2 record markers preceding each actual data record of 891 4-byte words.

Figure 1: IRIS Record Blocks

Word No.	Object	Comment
1	1 st Record Marker	'00000df4'x (swap B ₀ ,B ₁ = 3572 bytes)
2	2 nd Record Marker	'00000df0'x (swap B ₀ ,B ₁ = 3568 bytes)
3-893	Data Record	First word identifies record type (1 to 8)

Record blocks repeat multiple times. For contents of the data records, see section 3.1.

2.3 Key Science Data Fields

The primary science data fields in this data product is the calibrated radiances in units $W/cm^2/sr/cm^{-1}$ measured from wavenumber 400 to 1600 cm^{-1} .

Figure 2: Typical data coverage for a Nimbus 4 IRIS daily file. Each orbit for the day is color coded and labeled at the ascending equator crossing point.





Figure 3: Example of an IRIS spectrum for May 5, 1970 at 12:00:07 (orbit 365) over the Sahara.



3. Data Contents

The granularity of this data collection is daily. The first orbit may begin at the end of the preceding day, and the last orbit may end on the following day. Some data files may have corrupted data record types (see the appendix).

3.1 Data Records

There are discrepancies between the archived data product and what is described in the Nimbus-4 User’s Guide. The biggest difference is the elimination of record type 9, the summary record (User’s Guide table 4-12), which was (mostly) merged into record type 1, the documentation record (User’s Guide table 4-4), see Table 3-1-1 below. Another change is the wavenumber range is from about 400 cm⁻¹ to about 1600 cm⁻¹ (User’s Guide table 4-4 specifies 200 cm⁻¹ and 1660 cm⁻¹ at words 12 and 13), resulting in data arrays of size 862 instead of 1051, refer to tables 3-1-2 through 3-1-8. Another change is the orbit number field (4-byte integer) in the User’s Guide is changed to an orbit range (two 2-byte integers). The Nimbus 4 User’s Guide IRIS file format description seems to be copied from the Nimbus 3 User’s Guide IRIS file format description, apparently the format was changed at a later date, from orbit files to daily files.

Table 3-1-1: Record Type 1: Documentation/Summary Record

Word	Field Name	Units	Type	Comments
1	Record Type	-	I*4	1
2	Satellite ID	-	I*4	4
3	Initial Wave Number	cm ⁻¹	R*4	~400 cm ⁻¹
4	Final Wave Number	cm ⁻¹	R*4	~1600 cm ⁻¹
5	Wave Number Increment	-	R*4	1.39052 cm ⁻¹
6	Orbit Number Range	-	2 x I*2	Readout Orbit No.
7	Unknown Integer	?	I*4	?
8	Mean of Bolometer Temperature	K	R*4	
9	Standard Deviation of Bolometer Temperature	K	R*4	
10	Mean of Blackbody Temperature	K	R*4	
11	Standard Deviation of Blackbody Temperature	K	R*4	

12	Mean of Beamsplitter Temperature	K	R*4	
13	Standard Deviation of Beamsplitter Temperature	K	R*4	
14	Mean of Mirror Drive Motor Temperature	K	R*4	
15	Standard Deviation of Mirror Drive Motor Temperature	K	R*4	
16	Mean of IMCC Temperature	K	R*4	
17	Standard Deviation of IMCC Temperature	K	R*4	
18	Mean of Cooling Surface Temperature	K	R*4	
19	Standard Deviation of Cooling Surface Temperature	K	R*4	
20	Unknown Float	?	R*4	?
21	Unknown Float	?	R*4	?
22	Unknown Integer	?	I*4	?
23	Number of Reference Calibration Spectra	-	R*4	
24	Unknown Float	?	R*4	?
25	Number of Orbits	-	I*4	Number of Orbits in File
8(n-1)+26	Day	Days	I*4	Begin Day/Time of Orbit _n (n=1-18)
8(n-1)+27	Hour	Hours	I*4	
8(n-1)+28	Minute	Minutes	I*4	
8(n-1)+29	Second	Seconds	I*4	
8(n-1)+30	Day	Days	I*4	End Day/Time of Orbit _n (n=1-18)
8(n-1)+31	Hour	Hours	I*4	
8(n-1)+32	Minute	Minutes	I*4	
8(n-1)+33	Second	Seconds	I*4	
170-891	Spares		R*4	

Table 3-1-2: Record Type 2: Cold Reference Calibration Spectra Record

Word	Field Name	Units	Type	Comments
1	Record Type	-	I*4	2
2	Orbit Number Range	-	2 x I*2	Readout Orbit No.
3	Number of Cold Reference Calibration Spectra	-	I*4	
4	Average of Interferogram Peak Values	-	R*4	
5	Standard Deviation of Peak Values	-	R*4	
6	Average Position of Interferogram Peak Values	-	R*4	
7	Standard Deviation of Position of Peak Values	-	R*4	
8-29	Spares			
30-891	Averaged Cold Reference Spectrum Intensity	counts	R*4	

Table 3-1-3: Record Type 3: Warm Reference Calibration Spectra Record

Word	Field Name	Units	Type	Comments
1	Record Type	-	I*4	3
2	Orbit Number Range	-	2 x I*2	Readout Orbit No.
3	Number of Warm Reference Calibration Spectra	-	I*4	
4	Average of Interferogram Peak Values	-	R*4	
5	Standard Deviation of Peak Values	-	R*4	
6	Average Position of Interferogram Peak Values	-	R*4	
7	Standard Deviation of Position of Peak Values	-	R*4	
8-29	Spares			
30-891	Averaged Warm Reference Spectrum Intensity	counts	R*4	

Table 3-1-4: Record Type 4: Average Responsivity

Word	Field Name	Units	Type	Comments
1	Record Type	-	I*4	4
2	Orbit Number Range	-	2 x I*2	Readout Orbit No.
3-29	Spares			
30-891	Average responsivity	(cm ² ·sr·cm ⁻¹)/W	R*4	

Table 3-1-5: Record Type 5: Noise Equivalent Radiance

Word	Field Name	Units	Type	Comments
1	Record Type	-	I*4	4
2	Orbit Number Range	-	2 x I*2	Readout Orbit No.
3-29	Spares			
30-891	Noise Equivalent Radiance (NER)	(cm ² ·sr·cm ⁻¹)/W	R*4	

Table 3-1-6: Record Type 6: Average Instrument Temperature

Word	Field Name	Units	Type	Comments
1	Record Type	-	I*4	4
2	Orbit Number Range	-	2 x I*2	Readout Orbit No.
3-29	Spares			
30-891	Average Instrument Temperature	K	R*4	

Table 3-1-7: Record Type 7: Standard Deviation of the Instrument Temperature

Word	Field Name	Units	Type	Comments
1	Record Type	-	I*4	4
2	Orbit Number Range	-	2 x I*2	Readout Orbit No.
3-29	Spares			
30-891	Standard Deviation of the Instrument Temperature	K	R*4	

Table 3-1-8: Record Type 8: Calibrated Atmospheric Spectrum

Word	Field Name	Units	Type	Comments
1	Record Type	-	I*4	8
2	Orbit Number	-	I*4	Readout Orbit No.
3	Spectrum Number	-	I*4	Sequential within an orbit
4	Day	days	I*4	Day/Time of Spectrum
5	Hour	hours	I*4	
6	Minute	minutes	I*4	
7	Second	seconds	I*4	
8	Latitude	degrees	R*4	(-90 to +90)
9	Longitude	degrees	R*4	(0 to 360 westward)
10	Height of Satellite	km	R*4	
11	Solar Elevation Angle	Degrees	R*4	Listed as Bolometer Temp
12	Bolometer Temperature	K	R*4	No Redundant Sensor?
13	Blackbody Temperature	K	R*4	
14	Blackbody Temperature from Redundant Sensor	K	R*4	
15	Beamsplitter Temperature	K	R*4	
16	Temperature of Michelson Mirror Motor Drive	K	R*4	
17	IMCC Temperature	K	R*4	
18	Temperature of Cooling Surface	K	R*4	
19	IMCC Position	-	I*4	0 = warm ref. 2 = Earth 3 = cold ref.
20	+0.6 Volt Calibration	-	R*4	
21	0.0 Volt Calibration	-	R*4	
22	-0.6 Volt Calibration	-	R*4	
23	Calibration Transducer	-	R*4	
24	Unknown Float		R*4	Listed as Solar Elevation Angle

25	Spare		R*4	Listed as Word 26 in User's Guide
26	Number of Sync Bit Errors	-	R*4	Listed as Word 25 in User's Guide
27	Number of Gain Pulses Outside Center	-	R*4	
28	Time Indicator	-	I*4	0 = from raw tape 1 = computed
29	Spare			
30-891	Specific Intensity	W/(cm ² ·sr·cm ⁻¹)	R*4	

3.2 Metadata

The metadata are contained in a separate XML formatted file having the same name as the data file with .xml appended to it.

Table 3-2: Metadata attributes associated with the data file.

Name	Description
LongName	Long name of the data product.
ShortName	Short name of the data product.
VersionID	Product or collection version.
GranuleID	Granule identifier, i.e. the name of the file.
Format	File format of the data file.
ChecksumType	Type of checksum used.
ChecksumValue	The value of the calculated checksum.
SizeBytesDataGranule	Size of the file or granule in bytes.
InsertDateTime	Date and time when the granule was inserted into the archive. The format for date is YYYY-MM-DD and time is hh-mm-ss.
ProductionDateTime	Date and time the file was produced in format YYYY-MM-DDThh:mm:ss.sssssZ
RangeBeginningDate	Begin date when the data was collected in YYYY-MM-DD format.
RangeBeginningTime	Begin time of the date when the data was collected in hh-mm-ss format.
RangeEndingDate	End date when the data was collected in YYYY-MM-DD format.
RangeEndingTime	End time of the date when the data was collected in hh-mm-ss format.
WestBounding Coordinate	The westernmost longitude of the bounding rectangle(-180.0 to +180.0)
NorthBounding Coordinate	The northernmost latitude of the bounding rectangle(-90.0 to +90.0)
EastBounding Coordinate	The easternmost longitude of the bounding rectangle(-180.0 to +180.0)
SouthBounding Coordinate	The southernmost latitude of the bounding rectangle (-90.0 to +90.0)

PlatformShortName	Short name or acronym of the platform or satellite
InstrumentShortName	Short name or acronym of the instrument
SensorShortName	Short name or acronym of the sensor
FirstOrbitNumber	Number of the first orbit in the file
LastOrbitNumber	Number of the last orbit in the file
TapeFileName	Sequence number of file from the original tape preceded with 'f' and ending in '.dat'

4. Reading the Data

The data are written in a binary record-oriented format. Using the record format specification in the section above, users can write software to read the data files. Please note that the data were originally written using a big-endian format, therefore users on little-endian machines will need to swap bytes for the words. Also, the floating point data were written using IBM 360 machines, and must be converted if reading on a machine that understands IEEE floats (integers are not affected).

A sample FORTRAN program is included in the Appendix section which will read in the data records. Additionally two FORTRAN functions are included to perform byte swapping and conversion from IBM float to IEEE float.

5. Data Services

5.1 Reverb

The GES DISC provides basic temporal and advanced (event) searches through the EOSDIS Reverb data search and download interface:

<http://reverb.echo.nasa.gov>

Reverb allows users the ability to search on keywords, spatial region, and time period on datasets archived and various data centers. It offers various download options that suit users with different preferences and different levels of technical skills. To search for the SCAMS data enter [GES DISC IRISN4RAD V001](#) into the keyword field.

5.2 FTP

The Nimbus data products are available for users to download directly using anonymous FTP:

ftp://acdisc.gsfc.nasa.gov/data/s4pa/Nimbus4_IRIS_Level1B/IRISN4RAD.001/

The data are organized in directories by year with subdirectories by day of year. README, User's Guide and other documentation are located under the doc directory.

$W^{-1}\cdot cm^2\cdot sr\cdot cm^{-1}$

6. More Information

6.1 Web Resources

For other Nimbus data products, please see the GES DISC's Nimbus heritage data web page at:

<http://disc.gsfc.nasa.gov/nimbus/>

To search for other related data, please visit NASA's Global Change Master Directory at:

<http://gcmd.nasa.gov>.

6.2 Point of Contact

Name: GES DISC Help Desk

URL: <http://disc.gsfc.nasa.gov/>

E-mail: gsfc-help-disc@lists.nasa.gov

Phone: 301-614-5224

Fax: 301-614-5228

Address: Goddard Earth Sciences Data and Information Services Center

Attn: Help Desk

Code 610.2

NASA Goddard Space Flight Center

Greenbelt, MD 20771, USA

6.3 References

D. H. Staelin, et al, "The Nimbus-6 User's Guide - Section 4: The Scanning Microwave Spectrometer (SCAMS) Experiment", NASA Goddard Space Flight Center, Feb. 1975, Pages 59-86

7. Appendices

Acknowledgements

The Nimbus data recovery task at the GES DISC is funded by NASA's Earth Science Data and Information System program.

Acronyms

EOS: Earth Observing System

ESDIS: Earth Science and Data Information System

GES DISC: Goddard Earth Sciences Data and Information Services Center

GSFC: Goddard Space Flight Center

IRIS: Infrared Interferometer Spectrometer

L1: Level-1 Data

NASA: National Aeronautics and Space Administration

Reverb: ECHO's Next Generation Metadata and Service Discovery Tool

QA: Quality Assessment

UT: Universal Time

Problem Files

The following files have some corrupted data records:

1. IRIS-Nimbus4_1970m0411t2325_o56-65.dat (only record types 1-6 available, rest of file seems to be corrupted)
2. IRIS-Nimbus4_1970m0516t2327_o522-535.dat (problem at block 1365, wrong block size markers, corrupted record type 8)
3. IRIS-Nimbus4_1970m0518t0428_o536-546.dat (only record types 1-6 available, rest of file seems to be corrupted)
4. IRIS-Nimbus4_1970m0525t2344_o645-653.dat (problem at block 1485, wrong first block size marker, data seems OK)
5. IRIS-Nimbus4_1970m0530t0023_o696-707.dat (only record types 1-6 available, rest of file seems to be corrupted, wrong block size markers from block 272)
6. IRIS-Nimbus4_1970m1026t0046_o2697-2707.dat (problems from block 2100, wrong block size markers, time stamps are OK, but data values are all zero)

FORTRAN Code

```
C-----
C ^NAME: READ_IRIS
C
C ^DESCRIPTION:
C   This program opens and reads a Nimbus-4 IRIS level-1 data file
C   and prints the contents of the file to the screen. Data files
C   consist of blocks of data typically containing 8 record types,
C   each of size 891 words with 2 preceding words. See the Nimbus-4
C   Users Guide Section 4 for the IRIS file specification.
C
C ^MAJOR VARIABLES:
C   FNAME - name of input file
C
C ^NOTES:
C   Some files may not contain all 8 record types.
C
C   Compile: gfortran -o READ_IRIS.EXE READ_IRIS.FOR
C
C ^ORGANIZATION: NASA/GSFC, Code 610.2
C
C ^AUTHOR: James Johnson
C
C ^ADDRESS: james.johnson@nasa.gov
C
C ^CREATED: March 3, 2015
C-----

      CHARACTER          FNAME*1024      ! Name of input file
      INTEGER*4          IMARK1          ! First block size marker
      INTEGER*4          IMARK2          ! Second block size marker
      INTEGER*4          IDREC(891)      ! Data record is 891 4-byte words
      INTEGER*4          ITYPE           ! Record type
      REAL*4             WNINIT,         ! First Wavenumber
&                       WNSTEP,         ! Increment Step of Wavenumber
&                       WNLAST,        ! Last Wavenumber
&                       WAVNUM(862)    ! Wavenumber Array

C   Get the name of the input data file to read
      PRINT *, 'Enter the name of the input file:'
      READ (5, '(A)') FNAME

C   Open the specified input file
      OPEN (UNIT=1, FILE=FNAME, STATUS='OLD', ACCESS='DIRECT',
&         FORM='UNFORMATTED', RECL=4*893, ERR=99, IOSTAT=IOS)

C   Initialize N (block number)
      N=1

C   Loop through the file reading all blocks of data
      DO

          READ (1, REC=N, IOSTAT=IOS, ERR=90) IMARK1, IMARK2, IDREC
```

```

C      Read the first 4-byte word marker, first 2 bytes should = 'F40D'x
C      Swap bytes 1 & 2 gives '0DF4'x = 3572; total bytes in record block
      IF (IMARK1 .NE. IAND(IMARK1, Z'F40D')) THEN
          PRINT '("RECORD ",I11,": IMARK1  =",X,Z8.8)', N, IMARK1
          STOP
      ENDIF

C      Read the second 4-byte word marker, first 2 bytes should = 'F00D'x
C      Swap bytes 1 & 2 gives '0DF0'x = 3568; remaining bytes in block
      IF (IMARK2 .NE. IAND(IMARK2, Z'F00D')) THEN
          PRINT '("RECORD ",I11,": IMARK2  =",X,Z8.8)', N, IMARK2
          STOP
      ENDIF

C      Print the data records according to record type
      ITYPE = I4SWAP(IDREC(1))

      IF (ITYPE .EQ. 1) THEN
          CALL PRREC1(IDREC, WNINIT, WNLAST, WNSTEP)
          DO I=1, 862
              WAVNUM(I) = (I-1)*WNSTEP + WNINIT
          END DO
      ELSE IF (ITYPE .EQ. 2) THEN
          CALL PRREC2(IDREC)
      ELSE IF (ITYPE .EQ. 3) THEN
          CALL PRREC2(IDREC)
      ELSE IF (ITYPE .EQ. 4) THEN
          CALL PRREC4(IDREC, WAVNUM)
      ELSE IF (ITYPE .EQ. 5) THEN
          CALL PRREC4(IDREC, WAVNUM)
      ELSE IF (ITYPE .EQ. 6) THEN
          CALL PRREC4(IDREC, WAVNUM)
      ELSE IF (ITYPE .EQ. 7) THEN
          CALL PRREC4(IDREC, WAVNUM)
      ELSE IF (ITYPE .EQ. 8) THEN
          CALL PRREC8(IDREC, WAVNUM)
      ELSE
          PRINT '("RECORD ",I11,": UNKNOWN TYPE ",I11)', N, ITYPE
          STOP
      END IF

      N=N+1

      END DO

C      Close the input file
90  CLOSE(1)
      GOTO 100

99  PRINT '("ERROR: OPENING FILE, IOSTAT: ",I6)', IOS

100 STOP
      END

```

```

C-----
C ^SUBROUTINE: PRREC1
C
C   This subroutine will print data record type 1 to the screen
C   Record type 1 = documentation record (note this includes the
C   obsolete record type 9 = summary record)
C-----

      SUBROUTINE PRREC1(IDREC, WNINIT, WNLAST, WNSTEP)

      INTEGER*4      IDREC(891)      ! Data record
      INTEGER*2      IORBIT(2),      ! Orbit number range
&      I2TMP(2)      ! 2 temporary 2-byte integers
      INTEGER*4      I4TMP,          ! Temporary 4-byte integer
&      DAY(2),       ! Orbit begin/end day
&      HH(2),        ! Orbit begin/end hour
&      MM(2),        ! Orbit begin/end minute
&      SS(2),        ! Orbit begin/end second
      EQUIVALENCE   (I4TMP, I2TMP)

C   Word 1
      I4TMP = I4SWAP(IDREC(1))
      PRINT '( "DOCUMENTATION AND SUMMARY" )'
      PRINT '( "RECORD TYPE           = ",X,I11)', I4TMP

C   Word 2
      I4TMP = I4SWAP(IDREC(2))
      PRINT '( "SATELLITE             = ",X,I11)', I4TMP

C   Word 3
      I4TMP = I4SWAP(IDREC(3))
      WNINIT = R4IBM(I4TMP)
      PRINT '( "FIRST WAVENUMBER      = ",G12.6)', WNINIT

C   Word 4
      I4TMP = I4SWAP(IDREC(4))
      WNLAST = R4IBM(I4TMP)
      PRINT '( "FINAL WAVENUMBER      = ",G12.6)', WNLAST

C   Word 5
      I4TMP = I4SWAP(IDREC(5))
      WNSTEP = R4IBM(I4TMP)
      PRINT '( "WAVENUMBER INCREMENT  = ",G12.6)', WNSTEP

C   Word 6
      I4TMP = I4SWAP(IDREC(6))
      IORBIT(1) = I2TMP(2)
      IORBIT(2) = I2TMP(1)
      PRINT '( "ORBIT RANGE           = ",2(X,I5)', IORBIT

C   Word 7
      I4TMP = I4SWAP(IDREC(7))
      PRINT '( "UNKNOWN INTEGER 1     = ",X,I11)', I4TMP

C   Word 8
      I4TMP = I4SWAP(IDREC(8))
      PRINT '( "AVG BOLOMETER TEMP     = ",G12.6)', R4IBM(I4TMP)

```

```

C      Word 9
      I4TMP = I4SWAP(IDREC(9))
      PRINT '("SDEV BOLOMETER TEMP          = ",G12.6)', R4IBM(I4TMP)

C      Word 10
      I4TMP = I4SWAP(IDREC(10))
      PRINT '("AVG BLACKBODY TEMP          = ",G12.6)', R4IBM(I4TMP)

C      Word 11
      I4TMP = I4SWAP(IDREC(11))
      PRINT '("SDEV BLACKBODY TEMP          = ",G12.6)', R4IBM(I4TMP)

C      Word 12
      I4TMP = I4SWAP(IDREC(12))
      PRINT '("AVG BEAMSPLITTER TEMP        = ",G12.6)', R4IBM(I4TMP)

C      Word 13
      I4TMP = I4SWAP(IDREC(13))
      PRINT '("SDEV BEAMSPLITTER TEMP        = ",G12.6)', R4IBM(I4TMP)

C      Word 14
      I4TMP = I4SWAP(IDREC(14))
      PRINT '("AVG MIRROR DRIVE TEMP         = ",G12.6)', R4IBM(I4TMP)

C      Word 15
      I4TMP = I4SWAP(IDREC(15))
      PRINT '("SDEV MIRROR DRIVE TEMP         = ",G12.6)', R4IBM(I4TMP)

C      Word 16
      I4TMP = I4SWAP(IDREC(16))
      PRINT '("AVG IMCC TEMP                  = ",G12.6)', R4IBM(I4TMP)

C      Word 17
      I4TMP = I4SWAP(IDREC(17))
      PRINT '("SDEV IMCC TEMP                  = ",G12.6)', R4IBM(I4TMP)

C      Word 18
      I4TMP = I4SWAP(IDREC(18))
      PRINT '("AVG COOLING SFC TEMP           = ",G12.6)', R4IBM(I4TMP)

C      Word 19
      I4TMP = I4SWAP(IDREC(19))
      PRINT '("SDEV COOLING SFC TEMP           = ",G12.6)', R4IBM(I4TMP)

C      Word 20
      I4TMP = I4SWAP(IDREC(20))
      PRINT '("UNKNOWN FLOAT 1                = ",G12.6)', R4IBM(I4TMP)

C      Word 21
      I4TMP = I4SWAP(IDREC(21))
      PRINT '("UNKNOWN FLOAT 2                = ",G12.6)', R4IBM(I4TMP)

C      Word 22
      I4TMP = I4SWAP(IDREC(22))
      PRINT '("UNKNOWN INTEGER 2              = ",X,I11)', I4TMP

```

```

C      Word 23
      I4TMP = I4SWAP(IDREC(23))
      PRINT '("NO. CALIBRATION SPECTRA = ",G12.6)', R4IBM(I4TMP)

C      Word 24
      I4TMP = I4SWAP(IDREC(24))
      PRINT '("UNKNOWN FLOAT 3 = ",G12.6)', R4IBM(I4TMP)

C      Word 25
      NORBS = I4SWAP(IDREC(25))
      PRINT '("NUMBER OF ORBITS = ",X,I11)', NORBS

C      Words 26-(4*NORBS+26)
      PRINT '("ORBIT:          BEGIN          END")'
      PRINT '("          DAY  HOUR   MIN   SEC   DAY  HOUR   MIN   SEC")'
      DO 10 I=1,NORBS

          N=8*(I-1)+26

          DAY(1) = I4SWAP(IDREC(N))
          HH(1)  = I4SWAP(IDREC(N+1))
          MM(1)  = I4SWAP(IDREC(N+2))
          SS(1)  = I4SWAP(IDREC(N+3))

          DAY(2) = I4SWAP(IDREC(N+4))
          HH(2)  = I4SWAP(IDREC(N+5))
          MM(2)  = I4SWAP(IDREC(N+6))
          SS(2)  = I4SWAP(IDREC(N+7))

          PRINT '(2X,I2,8(2X,I4))', I, DAY(1), HH(1), MM(1), SS(1),
&              DAY(2), HH(2), MM(2), SS(2)

10 CONTINUE

C      Remaining words are spares

      RETURN
      END

```

```

C-----
C ^SUBROUTINE: PRREC2
C
C   This subroutine will print data record types 2 & 3 to the screen
C   Record type 2 = cold reference calibration spectra
C   Record type 3 = warm reference calibration spectra
C-----

      SUBROUTINE PRREC2(IDREC)

      INTEGER*4      IDREC(891)      ! Data record
      INTEGER*2      IORBIT(2),      ! Orbit number range
&      I2TMP(2)      ! 2 temporary 2-byte integers
      INTEGER*4      I4TMP           ! Temporary 4-byte integer
&      NSPECT(2)     ! Number of calibration spectra
      EQUIVALENCE   (I4TMP, I2TMP)

C   Word 1
      I4TMP = I4SWAP(IDREC(1))
      IF (I4TMP .EQ. 2) THEN
        PRINT '( /, "COLD REFERENCE CALIBRATION SPECTRA" )'
      ELSE
        PRINT '( /, "WARM REFERENCE CALIBRATION SPECTRA" )'
      ENDIF
      PRINT '( "RECORD TYPE           = ",X,I11)', I4TMP

C   Word 2
      I4TMP = I4SWAP(IDREC(2))
      IORBIT(1) = I2TMP(2)
      IORBIT(2) = I2TMP(1)
      PRINT '( "ORBIT RANGE           = ",2(X,I5)', IORBIT

C   Word 3
      NSPECT = I4SWAP(IDREC(3))
      PRINT '( "NUMBER OF SPECTRA     = ",X,I11)', NSPECT

C   Word 4
      I4TMP = I4SWAP(IDREC(4))
      PRINT '( "AVG PEAK VALUE         = ",G12.6)', R4IBM(I4TMP)

C   Word 5
      I4TMP = I4SWAP(IDREC(5))
      PRINT '( "SDEV PEAK VALUE       = ",G12.6)', R4IBM(I4TMP)

C   Word 6
      I4TMP = I4SWAP(IDREC(6))
      PRINT '( "AVG POSITION PEAK VALUE = ",G12.6)', R4IBM(I4TMP)

C   Word 7
      I4TMP = I4SWAP(IDREC(7))
      PRINT '( "SDEV POSITION PEAK VALUE = ",G12.6)', R4IBM(I4TMP)

C   Word 8 - 29 = spares

```

```
C      Words 30 - (NSPECT+29)
      PRINT '("DATA:",/, "      NUMBER      VALUE")'
      DO 10 I=1,NSPECT
          I4TMP = I4SWAP(IDREC(I-1+30))
          PRINT '(2X,I8,4X,G12.6)', I, R4IBM(I4TMP)
10 CONTINUE

C      Remaining words are spares

      RETURN
      END
```

```

C-----
C ^SUBROUTINE: PRREC4
C
C   This subroutine will print data record types 4-7 to the screen
C   Record type 4 = average responsivity
C   Record type 5 = noise equivalent radiance (NER)
C   Record type 6 = average of instrument temperature
C   Record type 7 = std dev of instrument temperature
C-----

      SUBROUTINE PRREC4(IDREC, WAVNUM)

      INTEGER*4      IDREC(891)      ! Data record
      REAL*4         WAVNUM(862)     ! Array of wavenumbers
      INTEGER*2      IORBIT(2),      ! Orbit number range
&      I2TMP(2)      ! 2 temporary 2-byte integers
      INTEGER*4      I4TMP           ! Temporary 4-byte integer
&      NSPECT(2)    ! Number of calibration spectra
      EQUIVALENCE   (I4TMP, I2TMP)

C   Word 1
      I4TMP = I4SWAP(IDREC(1))
      IF (I4TMP .EQ. 4) THEN
        PRINT '(/,"AVERAGE RESPONSIVITY")'
      ELSE IF (I4TMP .EQ. 5) THEN
        PRINT '(/,"NOISE EQUIVALENT RADIANCE (NER)")'
      ELSE IF (I4TMP .EQ. 6) THEN
        PRINT '(/,"AVERAGE OF INSTRUMENT TEMPERATURE")'
      ELSE IF (I4TMP .EQ. 7) THEN
        PRINT '(/,"STD DEV OF INSTRUMENT TEMPERATURE")'
      ELSE
        PRINT '(/,"UNKNOWN RECORD")'
        STOP
      ENDIF
      PRINT '("RECORD TYPE           = ",X,I11)', I4TMP

C   Word 2
      I4TMP = I4SWAP(IDREC(2))
      IORBIT(1) = I2TMP(2)
      IORBIT(2) = I2TMP(1)
      PRINT '("ORBIT RANGE           = ",2(X,I5))', IORBIT

C   Word 3 - 29 = spares

C   Words 30 - 891 Data Values at each Wavenumber
      PRINT '("DATA:",/," WAVENUMBER      VALUE")'
      DO 10 I=30,891
        I4TMP = I4SWAP(IDREC(I))
        PRINT '(2X,F8.2,4X,G12.6)', WAVNUM(I-29), R4IBM(I4TMP)
10 CONTINUE

C   Remaining words are spares

      RETURN
      END

```

```

C-----
C ^SUBROUTINE: PRREC8
C
C   This subroutine will print data record type 8 to the screen
C   Record type 8 = calibrated atmospheric spectrum
C-----

      SUBROUTINE PRREC8(IDREC, WAVNUM)

      INTEGER*4      IDREC(891)      ! Data record
      REAL*4         WAVNUM(862)     ! Array of wavenumbers
      INTEGER*4      I4TMP           ! Temporary 4-byte integer

C   Word 1
      I4TMP = I4SWAP(IDREC(1))
      PRINT '(/,"CALIBRATED ATMOSPHERIC SPECTRUM")'
      PRINT '("RECORD TYPE                = ",X,I11)', I4TMP

C   Word 2
      I4TMP = I4SWAP(IDREC(2))
      PRINT '("ORBIT NUMBER                = ",X,I11)', I4TMP

C   Word 3
      I4TMP = I4SWAP(IDREC(3))
      PRINT '("SPECTRUM NUMBER            = ",X,I11)', I4TMP

C   Word 4
      I4TMP = I4SWAP(IDREC(4))
      PRINT '("DAY NUMBER                  = ",X,I11)', I4TMP

C   Word 5
      I4TMP = I4SWAP(IDREC(5))
      PRINT '("HOUR                       = ",X,I11)', I4TMP

C   Word 6
      I4TMP = I4SWAP(IDREC(6))
      PRINT '("MINUTE                      = ",X,I11)', I4TMP

C   Word 7
      I4TMP = I4SWAP(IDREC(7))
      PRINT '("SECOND                     = ",X,I11)', I4TMP

C   Word 8
      I4TMP = I4SWAP(IDREC(8))
      PRINT '("LATITUDE                    = ",G12.6)', R4IBM(I4TMP)

C   Word 9
      I4TMP = I4SWAP(IDREC(9))
      PRINT '("LONGITUDE                   = ",G12.6)', R4IBM(I4TMP)

C   Word 10
      I4TMP = I4SWAP(IDREC(10))
      PRINT '("ALTITUDE                    = ",G12.6)', R4IBM(I4TMP)

C   Word 11
      I4TMP = I4SWAP(IDREC(11))
      PRINT '("SOLAR ELEVATION ANGLE       = ",G12.6)', R4IBM(I4TMP)

```

```

C      Word 12
      I4TMP = I4SWAP(IDREC(12))
      PRINT '("BOLOMETER TEMP                = ",G12.6)', R4IBM(I4TMP)

C      Word 13
      I4TMP = I4SWAP(IDREC(13))
      PRINT '("BLACKBODY TEMP                = ",G12.6)', R4IBM(I4TMP)

C      Word 14
      I4TMP = I4SWAP(IDREC(14))
      PRINT '("BLACKBODY TEMP REDUNDANT = ",G12.6)', R4IBM(I4TMP)

C      Word 15
      I4TMP = I4SWAP(IDREC(15))
      PRINT '("BEAMSPLITTER TEMPERATURE = ",G12.6)', R4IBM(I4TMP)

C      Word 16
      I4TMP = I4SWAP(IDREC(16))
      PRINT '("MIRROR MOTOR TEMPERATURE = ",G12.6)', R4IBM(I4TMP)

C      Word 17
      I4TMP = I4SWAP(IDREC(17))
      PRINT '("IMCC TEMPERATURE            = ",G12.6)', R4IBM(I4TMP)

C      Word 18
      I4TMP = I4SWAP(IDREC(18))
      PRINT '("COOLING SURFACE TEMP        = ",G12.6)', R4IBM(I4TMP)

C      Word 19
      I4TMP = I4SWAP(IDREC(19))
      PRINT '("IMCC POSITION                  = ",X,I11)', I4TMP

C      Word 20
      I4TMP = I4SWAP(IDREC(20))
      PRINT '("+0.6 VOLT CALIBRATION        = ",G12.6)', R4IBM(I4TMP)

C      Word 21
      I4TMP = I4SWAP(IDREC(21))
      PRINT '(" 0.0 VOLT CALIBRATION        = ",G12.6)', R4IBM(I4TMP)

C      Word 22
      I4TMP = I4SWAP(IDREC(22))
      PRINT '("-0.6 VOLT CALIBRATION        = ",G12.6)', R4IBM(I4TMP)

C      Word 23
      I4TMP = I4SWAP(IDREC(23))
      PRINT '("CALIBRATION TRANSDUCER      = ",G12.6)', R4IBM(I4TMP)

C      Word 24
      I4TMP = I4SWAP(IDREC(24))
      PRINT '("UNKNOWN FLOAT                    = ",G12.6)', R4IBM(I4TMP)

C      Word 25
      I4TMP = I4SWAP(IDREC(25))
      PRINT '("SPARE                          = ",G12.6)', R4IBM(I4TMP)

```

```

C      Word 26
      I4TMP = I4SWAP(IDREC(26))
      PRINT '("NO. SYNC BIT ERRORS      = ",G12.6)', R4IBM(I4TMP)

C      Word 27
      I4TMP = I4SWAP(IDREC(27))
      PRINT '("NO. GAIN PULSES OUTSIDE = ",G12.6)', R4IBM(I4TMP)

C      Word 28
      I4TMP = I4SWAP(IDREC(28))
      PRINT '("TIME INDICATOR          = ",X,I11)', I4TMP

C      Word 29 = spare

C      Words 30 - 891 Radiances at each wavelength
      PRINT '("DATA:",/, " WAVENUMBER    VALUE")'
      DO 10 I=30,891
         I4TMP = I4SWAP(IDREC(I))
         PRINT '(2X,F8.2,4X,G12.6)', WAVNUM(I-29), R4IBM(I4TMP)
10 CONTINUE

      RETURN
      END

```

```

C-----
C ^FUNCTION: I4SWAP
C
C   This subroutine will swap the bytes of a data element
C-----

      FUNCTION I4SWAP(INPUT)

      INTEGER*4      IWORD      ! Input 4-byte word
      INTEGER*4      IDROW      ! Byte-swapped 4-byte word
      CHARACTER      DATBUF(4)  ! Input data buffer
      CHARACTER      SWPBUF(4)  ! Output swapped buffer
      EQUIVALENCE    (IWORD, DATBUF)
      EQUIVALENCE    (IDROW, SWPBUF)

      IWORD = INPUT
      DO 10 K=1,4
         SWPBUF(K) = DATBUF(4-K+1)
10 CONTINUE
      I4SWAP = IDROW

      RETURN
      END

```

```

C-----
C ^FUNCTION: R4IBM
C
C   This function will convert an input word to an IBM float
C-----

      FUNCTION R4IBM(IWORD)

      INTEGER*4      IDROW          ! reverse the bits of input word
      REAL*8         A /16.0/       ! base number
      INTEGER*4      B /64/         ! exponent offset
      REAL*8         C /0.0/        ! fraction offset
      INTEGER*1      S              ! sign flag
      INTEGER*2      E              ! binary exponent
      REAL*8         F              ! binary fraction
      REAL*8         M              ! mantissa
      REAL*8         V              ! float value
      INTEGER*4      I              ! counter

      S = ISHFT(IWORD, -31)

      E = 0
      DO 10 I=0,6
         E = E + IAND(ISHFT(IWORD,-24),ISHFT(1,I))
      10 END DO

      IDROW = 0
      DO 11 I=0,31
         IF (IAND(IWORD,ISHFT(1,I)) .NE. 0) THEN
            IDROW = IOR(IDROW,(ISHFT(1,31-I)))
         END IF
      11 END DO

      F = 0.0
      DO 12 I=0,31
         IF (ISHFT(IAND(ISHFT(IDROW,-8),ISHFT(1,I)),1) .NE. 0) THEN
            F = F + 1 / FLOAT(ISHFT(IAND(ISHFT(IDROW,-8),ISHFT(1,I)),1))
         END IF
      12 END DO

      M = C + F                      ! calculate the mantissa
      V = (-1)**S * M * A**(E - B)    ! calculate the float value
      IF (ABS(V) .LT. 2.0**(-149)) THEN
         V = (-1)**S * 0.0           ! avoid underflow
      END IF

      R4IBM = V
      RETURN
      END

```